Ряды, интегралы, интегральные преобразования

1. Сумма и произведение ряда.
2. Интегрирование.
3. Интегральные преобразования.

Всюду далее примеры работы пакета Maple будут приведены в режиме интерфейса Worksheet Mode с текстовым режимом ввода команд и выражений (Text Mode). При выполнении заданий и упражнений используйте режим интерфейса Worksheet Mode с режимом ввода «двумерной математики» Maple Math Mode или ввода в строку Text Mode.

§1. Интегрирование функции одной переменной

Аналитическое и численное интегрирование

Неопределенный интеграл $\int f(x)dx$ вычисляется с помощью двух команд:

1) команды прямого исполнения – **`int(f, x)`**, где **`f`** – подынтегральная функция, **`x`** – переменная интегрирования; команду прямого исполнения **`int(f, x)`** можно записать с помощью шаблона $\int f\,\mathrm{d}x$ на палитре Expressions.

2) команды отложенного исполнения – **`Int(f, x)`** – где параметры команды такие же, как и в команде прямого исполнения **`int`**. Команда **`Int`** выдает на экран интеграл в аналитическом виде математической формулы.

Для вычисления определенного интеграла $\int_a^b f(x)dx$ в командах **`int`** и **`Int`** добавляются пределы интегрирования. Команду прямого исполнения для вычисления определенного интеграла **`int(f, x=a..b)`** можно записать с помощью шаблона $\int_a^b f\,\mathrm{d}x$ на палитре Expressions.

Рассмотрим пример:

```
> Int((1+cos(x))^2, x=0..Pi)=
 int((1+cos(x))^2, x=0..Pi);
```

$$\int_0^{\pi}(1+\cos(x))^2dx=\frac{3}{2}\pi$$

Если в команде интегрирования добавить опцию **`continuous`**: **`int(f, x, continuous)`**, то *Maple* будет игнорировать любые возможные разрывы подынтегральной функции в диапазоне интегрирования. Это позволяет вычислять несобственные интегралы от неограниченных функций. Несобственные интегралы с бесконечными пределами интегрирования вычисляются, если в параметрах команды **`int`** указывать, например, **`x=0..+infinity`**.

Численное интегрирование выполняется командой **`evalf(int(f, x=x1..x2), e)`**, где **`e`** – точность вычислений (число знаков после запятой).

Интегралы, зависящие от параметра. Ограничения для параметров

Если требуется вычислить интеграл, зависящий от параметра, то его значение может зависеть от знака этого параметра или каких-либо других ограничений. Рассмотрим в качестве примера интеграл $\int_0^{+\infty} e^{-ax}dx$, который, как известно из математического анализа, сходится при $a>0$ и расходится при $a<0$. Если вычислить его сразу, то получится:

```
> Int(exp(-a*x),x=0..+infinity)=
  int(exp(-a*x),x=0..+infinity);
```

$$\int_0^{+\infty} e^{(-ax)}dx = \lim_{x\to\infty} -\frac{e^{(-ax)}-1}{a}.$$

Для получения явного аналитического результата вычислений следует сделать какие-либо предположения о значении параметров, то есть наложить на них ограничения. Это можно сделать при помощи команды **`assume(expr1)`**, где **`expr1`** – неравенство. Дополнительные ограничения вводятся с помощью команды **`additionally(expr2)`**, где **`expr2`** – другое неравенство, ограничивающее значение параметра с другой стороны.

После наложения ограничений на параметр *Maple* добавляет к его имени символ (~), например параметр **`a`**, на который были наложены некоторые ограничения, в сроке вывода будет иметь вид: a~.

Описание наложенных ограничений параметра **`a`** можно вызвать командой **`about(a)`**. Пример: наложить ограничения на параметр a такие, что $a>-1$, $a\le 3$:

```
> assume(a>-1); additionally(a<=3);
> about(a);
Originally a, renamed a~:
  is assumed to be: RealRange(Open(-1),3)
```

Вернемся к вычислению интеграла с параметром $\int_0^{+\infty} e^{-ax}dx$, которое следует производить в таком порядке:

```
> assume(a>0);
> Int(exp(-a*x),x=0..+infinity)=
  int(exp(-a*x),x=0..+infinity);
```

$$\int_0^{+\infty} e^{(-a\sim x)}dx = \frac{1}{a\sim}$$

Обучение основным методам интегрирования

В *Maple* имеется пакет **`student`**, предназначенный для обучения математике. Он содержит набор подпрограмм, предназначенных для выполнения расчетов шаг за шагом, так, чтобы была понятна последовательность действий, приводящих к результату. К таким командам относятся интегрирование по частям **`inparts`** и замена переменной **`changevar`**.

Формула интегрирования по частям:

$$\int u(x)v'(x)dx = u(x)v(x) - \int u'(x)v(x)dx$$

Если обозначить подынтегральную функцию $f=u(x)v'(x)$, то параметры команды интегрирования по частям такие: **`intparts(Int(f, x), u)`**, где **`u`** – именно та функция **`u(x)`**, производную от которой предстоит вычислить по формуле интегрирования по частям.

Если в интеграле требуется сделать замену переменных $x=g(t)$ или $t=h(x)$, то параметры команды замены переменных такие: **`changevar(h(x)=t, Int(f, x), t),`** где **`t`** – новая переменная.

Обе команды **`intparts`** и **`changevar`** не вычисляют окончательно интеграл, а лишь производят промежуточную выкладку. Для того, чтобы получить окончательный ответ, следует, после выполнения этих команд ввести команду **`value(%);`** где `%` - обозначают предыдущую строку.

Не забудьте, перед использованием описанных здесь команд обязательно загрузить пакет **`student`** командой **`with(student).`**

Задание 1

1. Найти неопределенные интегралы: а) $\int \cos x \cos 2x \cos 3x dx$;

б) $\int \frac{3x^4+4}{x^2(x^2+1)^3}dx$.

```
> Int(cos(x)*cos(2*x)*cos(3*x),x)=
  int(cos(x)*cos(2*x)*cos(3*x), x);
```

$$\int \cos(x)\cos(2x)\cos(3x)dx = \frac{1}{8}\sin(2x)+\frac{1}{16}\sin(4x)+\frac{1}{24}\sin(6x)+\frac{1}{4}x$$

```
> Int((3*x^4+4)/(x^2*(x^2+1)^3),x)=
  int((3*x^4+4)/(x^2*(x^2+1)^3),x);
```

$$\int \frac{3x^4+4}{x^2\left(x^2+1\right)^3}\,dx = -\frac{25}{8}\frac{x}{x^2+1}-\frac{57}{8}\arctan(x)-\frac{4}{x}-\frac{7}{4}\frac{x}{\left(x^2+1\right)^2}$$

2. Найти определенный интеграл $\int_0^{\pi/2} \frac{\sin x \cos x dx}{(a^2\cos^2 x+b^2\sin^2 x)^2}$, при условии $a>0$, $b>0$.

```
> assume (a>0,b>0);
> Int(sin(x)*cos(x)/(a^2*cos(x)^2+b^2*sin(x)^2)^2,x=0..Pi/2)=
simplify(int(sin(x)*cos(x)/(a^2*cos(x)^2+b^2*sin(x)^2)^2,x=0..
Pi/2));
```

$$\int_0^{\frac{1}{2}\pi} \frac{\sin(x)\,\cos(x)}{\left(a\sim^2\cos(x)^2+b\sim^2\sin(x)^2\right)^2}\,dx = \frac{1}{2\,a\sim^2\,b\sim^2}$$

3. Найти несобственный интеграл $\int_0^{+\infty} \frac{1-e^{-ax^2}}{xe^{x^2}}dx$, при $a>-1$

```
> restart; assume(a>-1);
> Int((1-exp(-a*x^2))/(x*exp(x^2)),
 x=0..+infinity)=int((1-exp(-a*x^2))/(x*exp(x^2)),
 x=0..+infinity);
```

$$\int_0^{\infty} \frac{1 - e^{-a\sim x^2}}{x\, e^{x^2}}\, dx = \frac{1}{2} \ln(1 + a\sim)$$

4. Численно найти интеграл $\int\limits_{\pi/6}^{\pi/4} \frac{\cos x}{x} dx$

```
> Int(cos(x)/x, x=Pi/6..Pi/4)=evalf(int(cos(x)/x,
 x=Pi/6..Pi/4), 15);
```

$$\int\limits_{\pi/6}^{\pi/4} \frac{\cos(x)}{x} dx = .322922981\ 113732$$

5. Полностью проделать все этапы вычисления интеграла $\int x^3 \sin x dx$ по частям.

```
> restart; with(student): J=Int(x^3*sin(x),x);
```

$$J = \int x^3 \sin(x) dx$$

```
> J=intparts(Int(x^3*sin(x),x),x^3);
```

$$J = -x^3 \cos(x) - \int -3x^2 \cos(x) dx$$

```
> intparts(%,x^2);
```

$$J = -x^3 \cos(x) + 3x^2 \sin(x) + \int -6x \sin(x) dx$$

```
> intparts(%,x);
```

$$J = -x^3 \cos(x) + 3x^2 \sin(x) + 6x \cos(x) - \int 6 \cos(x) dx$$

```
> value(%);
```

$$J = -x^3 \cos(x) + 3x^2 \sin(x) + 6x \cos(x) - 6 \sin(x)$$

6. Вычислить интеграл $\int\limits_{-\pi/2}^{\pi/2} \frac{dx}{1+\cos x}$ с помощью универсальной подстановки $\mathrm{tg}\frac{x}{2} = t$.

```
> J=Int(1/(1+cos(x)), x=-Pi/2..Pi/2);
```

$$J = \int\limits_{-\pi/2}^{\pi/2} \frac{1}{1+\cos(x)} dx$$

```
> J=changevar(tan(x/2)=t,Int(1/(1+cos(x)),
x=-Pi/2..Pi/2), t);
```

$$J = \int\limits_{-1}^{1} \frac{1}{(1+\cos(2\arctan(t)))(1+t^2)} dt$$

```
> value(%);
```

$$J=2$$

§2. Интегральное исчисление функций многих переменных, вычисление двойных и тройных интегралов

В *Maple* имеются две специальные команды для вычисления двойных и тройных интегралов, содержащиеся в библиотеке **student**.

Для вычисления двойных интегралов $\iint\limits_D f(x,y)dxdy$ используется команда **Doubleint(f(x, y), D)**, где **D** – область интегрирования, записываемая в одном из следующих форматов:

- **x=x1..x2, y=y1..y2**, где числа **x1, x2, y1, y2** задают прямоугольную область интегрирования;
- **x=f1(y)..f2(y), y=y1..y2**, где **f1(y), f2(y)** – линии, ограничивающие область интегрирования слева и справа на интервале от **y1** до **y2**;
- **x=x1..x2, y=g1(x)..g2(x)** , где **g1(y), g2(y)** – линии, ограничивающие область интегрирования снизу и сверху на интервале от **x1** до **x2**.

Для вычисления тройных интегралов $\iiint\limits_V f(x,y,z)dxdydz$ используется команда **Tripleint(f(x, y, z),x, y, z, V)**, где **V** – область интегрирования.

Обе эти команды являются командами отложенного действия. Чтобы получить значение интеграла, следует использовать команду **value(%)**.

Повторные интегралы можно вычислять с помощью вложения команды **int**, например, повторный интеграл $\int\limits_0^2 dy\int\limits_0^1 x^2y^3dx$ вычисляется командой

```
> int(int(x^2*y^3, x=0..1), y=0..2);
```

$$\frac{4}{3}$$

Задание 2

1. Вычислить повторный интеграл $\int\limits_2^4 dy\int\limits_0^y \frac{y^3}{x^2+y^2}dx$

```
> Int(Int(y^3/(x^2+y^2),x=0..y),y=2..4)=
  int(int(y^3/(x^2+y^2), x=0..y),y=2..4);
```

$$\int\limits_2^4 dy\int\limits_0^y \frac{y^3}{x^2+y^2}dx = \frac{14}{3}\pi$$

2. Вычислить двойной интеграл $\iint\limits_D \sin(x+2y)dxdy$ по области, ограниченной линиями $y=0,\ y=x,\ x+y=\frac{\pi}{2}$.

Замечание: сначала следует описать область интегрирования D в виде неравенств: $D=\{(x,y): y\le x\le \frac{\pi}{2}-y,\ 0\le y\le \frac{\pi}{2}\}$

```
> restart: with(student):
> J:=Doubleint(sin(x+2*y), x=y..Pi/2-y, y=0..Pi/2);
```

$$J := \int\limits_0^{\frac{1}{2}\pi}\int\limits_y^{\frac{1}{2}\pi-y} \sin(x+2y)dxdy$$

```
> J:=value(%);
```

$$J := \frac{2}{3}$$

3. Вычислить тройной интеграл $\int\limits_{-1}^{1} dx \int\limits_{x^2}^{1} dy \int\limits_{0}^{2} (4+z) dz$.

Замечание: следует помнить, что порядок интегрирования определяется последовательностью пределов, поэтому сначала внутренние указываются пределы, содержащие функции.

```
> J:=Tripleint(4+z, y=x^2..1,x=-1..1, z=0..2);
```

$$J := \int\limits_{0}^{2}\int\limits_{-1}^{1}\int\limits_{x^2}^{1} 4+z dydxdz$$

```
> J:=value(%);
```

$$J := \frac{40}{3}$$

§3. Сумма и произведение ряда

Вычисление суммы ряда и произведений

Конечные и бесконечные суммы $\sum\limits_{n=a}^{b} S(n)$ вычисляются командой прямого исполнения **sum** и отложенного исполнения **Sum**. Аргументы этих команд одинаковые: **sum(expr, n=a..b)**, где **expr** – выражение, зависящее от индекса суммирования, **a..b** – пределы индекса суммирования, указывающие, что суммировать следует от **n=a** до **n=b**.

Команду прямого исполнения **sum(f, i=k..n)** можно набрать с помощью шаблона $\sum\limits_{i=k}^{n} f$ на палитре Expressions.

Если требуется вычислить сумму бесконечного ряда, то в качестве верхнего предела вводится **infinity**.

Аналогичным образом вычисляются произведения $\prod\limits_{n=a}^{b} P(n)$ командами прямого **product(P(n),n=a..b)** и отложенного действий **Product P(n),n=a..b)**.

Команду прямого исполнения **product(f, i=k..n)** можно набрать с помощью шаблона $\prod\limits_{i=k}^{n} f$ на палитре Expressions.

Задание 3

1. Найти полную и *N*-частичную суммы ряда, общий член которого равен: $a_n = \frac{1}{(3n-2)(3n+1)}$.

```
> restart: a[n]:=1/((3*n-2)*(3*n+1));
> S[N]:=Sum(a[n], n=1..N)=sum(a[n], n=1..N);
> S:=limit(rhs(S[N]), N=+infinity);
> S:=sum(a[n], n=1..+infinity);
```

$$a_n := \frac{1}{(3\,n-2)\,(3\,n+1)}$$

$$S_N := \sum_{n=1}^{N} \frac{1}{(3n-2)(3n+1)} =$$

$$-\frac{1}{3(3N+1)} + \frac{1}{3}$$

$$S := \frac{1}{3}$$

$$S := \frac{1}{3}$$

2. К какой функции сходится степенной ряд: $\sum_{n=1}^{\infty}(-1)^{n+1}n^2x^n$?

```
> Sum((-1)^(n+1)*n^2*x^n, n=1..infinity)=
 sum((-1)^(n+1)*n^2*x^n, n=1..infinity);
```

$$\sum_{n=1}^{\infty} (-1)^{n+1} n^2 x^n = -\frac{x(x-1)}{(x+1)^3}$$

3. Найти сумму степенного ряда $\sum_{n=0}^{\infty}\frac{(1+x)^n}{(n+1)n!}$.

```
> Sum((1+x)^n/((n+1)*n!), n=0..infinity)=
 sum((1+x)^n/((n+1)*n!), n=0..infinity);
```

$$\sum_{n=0}^{\infty} \frac{(x+1)^n}{(n+1)\,n!} = \frac{e^{x+1}-1}{x+1}$$

4. Вычислить бесконечное произведение: $\prod_{n=2}^{\infty}\frac{n^3-1}{n^3+1}$

```
> Product((n^3-1)/(n^3+1),n=2..infinity)=
 product((n^3-1)/(n^3+1), n=2..infinity);
```

$$\prod_{n=2}^{\infty}\frac{n^3-1}{n^3+1} = \frac{2}{3}.$$

§4. Интегральные преобразования

В *Maple* имеется пакет **inttrans**, в котором содержатся команды различных интегральных преобразований.

Преобразование Фурье

Прямое преобразование Фурье функции $f(x)$ вычисляется по формуле

$$F(k) = \int_{-\infty}^{\infty} f(x)e^{-ikx}\,dx\,.$$

В *Maple* оно может быть найдено командой **fourier(f(x),x,k)**, где **x** – переменная, по которой производится преобразование, **k** – имя переменной, которое следует присвоить параметру преобразования.

Обратное преобразование Фурье задается формулой

$$f(x) = \frac{1}{2\pi}\int_{-\infty}^{\infty} F(k)e^{ikx}\,dk$$

и вычисляется командой **invfourier(F(k),k,x)**.

Описанное выше прямое и обратное преобразования Фурье называются комплексными и применяются в тех случаях, когда функция *f*(*x*) задана на всей числовой оси. Если функция *f*(*x*) задана только при *x*>0, то рекомендуется использовать синус- и косинус- преобразования Фурье.

Прямое и обратное синус-преобразования Фурье функции *f*(*x*) определяются формулами

$$F(k)=\sqrt{\frac{2}{\pi}}\int\limits_{0}^{\infty} f(x)\sin kx dx \text{ и } f(x)=\sqrt{\frac{2}{\pi}}\int\limits_{-\infty}^{\infty} F(k)\sin kx dk.$$

Поскольку формулы синус-преобразования Фурье симметричны относительно замены *x* на *k*, то в *Maple* эти преобразования вычисляются одной командой, но с различным порядком указания параметров: **`fouriersin(f(x),x,k)`** – вычисляет прямое синус-преобразование Фурье; **`fouriersin(F(k),k,x)`** – вычисляет обратное синус-преобразование Фурье.

Аналогично, прямое и обратное косинус-преобразования Фурье функции *f*(*x*) определяются формулами

$$F(k)=\sqrt{\frac{2}{\pi}}\int\limits_{0}^{\infty} f(x)\cos kx dx \text{ и } f(x)=\sqrt{\frac{2}{\pi}}\int\limits_{-\infty}^{\infty} F(k)\cos kx dk.$$

В *Maple* эти преобразования вычисляются одной командой, но с различным порядком указания параметров: **`fouriercos(f(x),x,k)`** – вычисляет прямое косинус-преобразование Фурье; **`fouriercos(F(k),k,x)`** – вычисляет обратное косинус-преобразование Фурье.

Задание 4.1

1. Для функции $f(x)=e^{-a|x|}$, $a>0$ найти преобразование Фурье.

```
> restart:with(inttrans): assume(a>0):
> fourier(exp(-a*abs(x)),x,k);
```

$$\frac{2\,a\sim}{a\sim^2+k^2}$$

2. Для функции $F(k)=\dfrac{1}{k^2-a^2}$, *a*>0 найти обратное преобразование Фурье.

```
> restart; with(inttrans): assume(a > 0); invfourier(1/(k^2-a^2), k, x);
```

$$\frac{1}{2}\,\frac{\sin(a\sim x)\,(-2\,\mathrm{Heaviside}(x)+1)}{a\sim}$$

После обратного преобразования Фурье результат часто содержит специальные функции. В данном примере в строке вывода появилась функция Хевисайда:

$$\mathrm{Heaviside}(x)=\begin{cases}1, x>0\\ 0, x<0\end{cases}$$

3. Для функции $f(x)=e^{-ax}\sin bx$, *a*>0 найти синус- и косинус- преобразования Фурье.

```
> f:=exp(-a*x)*sin(b*x):
> fouriercos(f,x,k);
```

$$\frac{\sqrt{2}\left(\frac{1}{2}\,\frac{k+b}{a\sim^2+(k+b)^2}+\frac{1}{2}\,\frac{b-k}{a\sim^2+(b-k)^2}\right)}{\sqrt{\pi}}$$

```
> fouriercos(f,x,k);
```

$$\frac{1}{2}\frac{\sqrt{2}a\sim\left(\frac{1}{a\sim^2+(b-k)^2}-\frac{1}{a\sim^2+(k+b)^2}\right)}{\sqrt{\pi}}$$

Преобразование Лапласа

Преобразование Лапласа функции $f(x)$ (если оно существует) вычисляется по формуле:

$$F(p)=\int_0^{\infty} f(x)e^{-px}dx\,.$$

Получаемая функция $F(p)$ называется изображением.

В *Maple* это преобразование вычисляется командой **`laplace(f(x),x,p)`** , где **`x`** – переменная, по которой производится преобразование, **`p`** – имя переменной, которое следует присвоить параметру преобразования.

Обратное преобразование Лапласа (называется оригиналом) вычисляется по формуле:

$$f(x)=\frac{1}{2\pi i}\int_{a-i\infty}^{a+i\infty} F(p)e^{px}dp\,.$$

Оригинал $f(x)$ (если он существует) может быть найден по изображению $F(p)$ командой **`invlaplace(F(p),p,x)`**.

Задание 4.2

1. Найти изображение функции $f(x)=\cos ax\,\mathrm{sh}\,bx$.

```
> restart:with(inttrans):
> F(p)=laplace(cos(a*x)*sinh(b*x), x, p);
```

$$F(p)=\frac{b\left(-a^2-b^2+p^2\right)}{\left((p+b)^2+a^2\right)\left((p-b)^2+a^2\right)}$$

2. Найти оригинал Лапласа функции $F(p)=\dfrac{1}{p^2+2ap}$, $a>0$.

```
> assume(a>0): invlaplace(1/(p^2+2*a*p),p,x):
> combine(%,trig);
```

$$\frac{1}{2}\frac{1-\mathrm{e}^{(-2a\sim x)}}{a\sim}\,.$$